



The Dyalog Cookbook

for Windows

Stephen Taylor & Kai Jaeger

The Dyalog Cookbook

Stephen Taylor and Kai Jaeger

© 2015 - 2016 Stephen Taylor and Kai Jaeger

With grateful thanks to Kenneth E. Iverson (1920-2004), for a language we can think in, and to his collaborators and followers who make it work.

Contents

Introduction	i
What you need to use the Dyalog Cookbook	ii
 1. Packaging your application	 1
Structure	2
How can you distribute your program?	2
Where should you keep the code?	3
Versions	3
The MyApp workspace	4
Project Gutenberg	7
MyApp reloaded	7
Building from a DYAPP	8
 Package MyApp as an executable	 12
Output to the session log	12
Reading arguments from the command line	12
 Logging what happens	 14
Where to keep the logfiles?	16
 Handling errors	 18
What are we missing?	18
Foreseen errors	18
Unforeseen errors	22
Discussion	26
Offcuts	26
 Initialising with INI files	 27
 Testing: the sound of breaking glass	 28
 Testing in different versions of Windows	 29
 User interface – native	 30

CONTENTS

A simple UI with native Dyalog forms	30
Providing help	31
Writing an installer	32
Working with other processes	33
Launching tasks	33
Running as a Windows service	33
Sharing files	33
Communicating through TCP/IP	33
Managing your source code	34
Documentation	34
FIRE	34
CompareSimple	34
Useful user commands	34
Storing data	35
Native and component files	35
XML	35
JSON	35
Relational databases	35
User interface – WPF	36
User interface – HTTP	37
Deploying as a Web application	38
Deploying as a Web service	39
 2. Professional programming practices	 40
Stay DRY – don't repeat yourself	41
APL.local	42
Defining ubiquitous utilities	42
Some utilities you might like to make ubiquitous	42
Hooray for arrays	43
Functional is funky	44
Complex data structures	45

CONTENTS

Passing parameters perfectly	46
The ghaſtlineſſ of globals	47
Coding with claſſ	48

Introduction

You want to write an application in Dyalog APL. You have already learned enough of the language to put some of your specialist knowledge into functions you can use in your work. The code works for you. Now you want to turn it into an application others can use. Perhaps even sell it.

This is where you need professional programming skills. How to install your code into an unknown computer. Have it catch, handle and log errors. Manage the different versions of your software as it evolves. Provide online help.

You are not a professional programmer. You don't have those skills. Perhaps you need a professional programmer to turn your code into an application. But you've come a long way already. Perhaps you can get there by yourself - with the Dyalog Cookbook.

The Dyalog Cookbook is about how to turn your Dyalog code into an application. We'll cover packaging your code into a robust environment. And we'll introduce some software development tools you've managed without so far, which will make your life easier.

You might continue as the sole developer of your application for a long time yet. But if it becomes a successful product you will eventually want other programmers collaborating on it. So we'll set up your code in a source-control system that will accommodate that. Even while you remain a sole developer, a source-control system will also allow you to roll back and recover from your own mistakes.

Not so long ago it was sufficient for an application to be packaged as an EXE that could be installed and run on other PCs. Nowadays many corporate clients run programs in terminal servers or in private clouds. So we'll look at how to organise your program to run as tasks that communicate with each other. Many applications written in Dyalog focus on some kind of numerical analysis, and can have CPU-intensive tasks. We'll look at how such tasks can be packaged to run either in background or on remote machines.



Many of these issues are entangled with each other. We'll arrive at the best solutions by way of some interim solutions, so you get to understand on the way the issues and how their solutions work. In Part 1 you will find 'framework' code for your application, growing more complex as the book progresses. You can find scripts for these interim versions on the book website at dyalog.com. Watch out: they are interim solutions. By all means adapt the last as a framework for your application, but do so knowing how it works!

Finally in Part 2 we'll introduce some professional writing techniques that might make maintaining your code easier – in what we hope will be a long useful future for it!

What you need to use the Dyalog Cookbook

- The Dyalog Version 14 Unicode interpreter or later.
- To know how to use namespaces, classes and instances. Much of the utility code in the Cookbook is packaged into classes. This is the form in which it is easiest for you to slide the code into your app without name conflicts. We recommend you use classes to organise your own code! But even if you don't, you need to know at least how to use classes. This is a deep subject, but all you need to know is the basics: creating an instance of a class and using its methods and properties, or just using the methods of a static class. See [Dyalog Programmer's Reference Guide](#) for an introduction.
- To know how to use SALT, the Simple APL Library Toolkit. Just the basics, including editing namespace and class scripts. See [Dyalog SALT Reference Guide](#).

Stephen Taylor & Kai Jaeger

1. Packaging your application

Structure

In this chapter we consider your choices for making your program available to others, and for taking care of the source code, including tracking the changes through successive versions.

To follow this, we'll make a very simple program. It counts the frequency of letters used in one or multiple text files. (This is simple, but useful in cryptanalysis, at least at hobby level.) We'll put the source code under version control, and package the program for use.

Let's assume you've done the convenient thing. Your code is in a workspace. Everything it needs to run is defined in the workspace. Maybe you set a latent expression, so the program starts when you load the workspace.

In this chapter, we shall convert a DWS (saved workspace) to some DYALOG scripts and a DYAPP script to assemble an active workspace from them.

How can you distribute your program?

Send a workspace file (DWS)

Could not be simpler. If your user has a Dyalog interpreter, she can also save and send you the crash workspace if your program hits an execution error. But she will also be able to read your code – which might be more than you wish for.

If she doesn't have an interpreter, and you are not worried about her someday getting one and reading your code, and you have a Run-Time Agreement with Dyalog, you can send her the Dyalog Run-Time DLL[^ext] with the workspace. The Run-Time DLL will not allow the program to suspend, so when the program breaks the task will vanish, and your user won't see your code. All right so far. But she will also not have a crash workspace to send you. So you need your program to catch and report any errors before it dies.

Send an executable file (EXE)

This is the simplest form of the program to install, because there is nothing else it needs to run: everything is embedded within the EXE. You export the workspace as an EXE, which can have the Dyalog Run-Time DLL bound into it. The code cannot be read. As with the workspace-based runtime above, your program cannot suspend, so you need it to catch and report any errors before dying.

We'll do that!

Where should you keep the code?

Let's start by considering the workspace you will export as an EXE.

The first point is PCs have a lot of memory relative to your application code volume. So all your Dyalog code will be in the workspace. That's probably where you have it right now: all saved in a workspace.

Your workspace is like your desk top – a great place to get work done, but a poor place to store things. In particular it does nothing to help you track changes and revert to an earlier version.

Sometimes a code change turns out to be for the worse, and you need to undo it. Perhaps the change you need to undo is not the most recent change.

We'll keep the program in manageable pieces – 'modules' – and keep those pieces in text files under version control.

For this there are many *source-control management* (SCM) systems and repositories available. Subversion, SourceForge, GitHub and Mercurial are presently popular. These SCMs support multiple programmers working on the same program, and have sophisticated features to help resolve conflicts between them.

Whichever SCM you use (we used GitHub for writing this book and the code in it) your source code will comprise class and namespace scripts (DYALOGs) and a *build script* (DYAPP) to assemble them.

You'll keep your local working copy (or *branch* in Git terms) in whatever folder you please. We'll refer to this *working folder* as `Z:\` but it will of course be wherever suits you.

Versions

In real life you will produce successive *versions* of your program, each better than the last. In an ideal world, all your users will have and use the latest version. In that ideal world, you have only one version to maintain: the latest. In the real world, your users will have and use multiple versions. If you charge for upgrading to a newer version, this will surely happen. And even in your ideal world, you have to maintain at least two versions: the latest and the next.

What does it mean to maintain a version? At the very minimum, you keep the source code for it, so you could recreate its EXE from scratch, exactly as it was distributed. There will be things you want to improve, and perhaps bugs you must fix. Those will all go into the next version, of course. But some you may need to put into the released version and re-issue it to current users as a patch.

So in *The Dyalog Cookbook* we shall develop in successive versions. Our 'versions' are not ready to ship, so are probably better considered as milestones on the way to version 1.0. You could think of them as versions 0.1, 0.2 and so on. But we'll just refer to them as Versions 1, 2, and so on.

Our first version won't even be ready to export as an EXE. It will just recreate MyApp.DWS from scripts: a DYAPP and some DYALOGs. We'll call it Version 0.

The MyApp workspace

We suppose you already have a workspace in which your program runs. We mean to get from your workspace to class and namespace scripts and a build script.

We don't have your wonderful code to hand so we'll use ours. We'll use a very small and simple program, so we can focus on packaging the code as an application, not on writing the application itself.

Your application will of course be much more interesting!

So we'll begin with the MyApp workspace. It's trivially simple (we'll extend a bit what it does as we go) but for now it will stand in for your much more interesting code.

On encryption

Frequency counting relies on the distribution of letters being more or less constant for any given language. It is the first step in breaking a substitution cypher. Substitution cyphers have been superseded by public-private key encryption, and are mainly of historical interest, or for studying cryptanalysis. But they are also fun to play with.

We recommend *The Code Book: The secret history of codes & code-breaking* by Simon Singh and *In Code* by Sarah Flannery as introductions if you find this subject interesting.

From the `code\v01` folder on the book website load `LetterCount.dws`. Again, this is just the stand-in for your own code. Here's a quick tour.

Discussion

Function `TxtToCsv` takes the filepath of a TXT (text file) and writes a sibling CSV file containing the frequency count for the letters in the file. It uses function `CountLetters` to produce the table.

`CountLetters` returns a table of the letters in `⌈A` and the number of times each is found in the text. The count is insensitive to case and ignores accents, mapping accented to unaccented characters:

```
ACCENTS
ÁÂÃÄÅÇÐÈÉÊËÌÍÎÏÑÒÓÔÕÖØÙÚÛÜÝ
AAAAAACDEEEIIIIINOOOOOUUUUY
```

That amounts to five functions. Two of them are specific to the application: `TxtToCsv` and `CountLetters`. The other three – `caseUp`, `join`, and `map` are utilities, of general use.

`caseUp` uses the fast case-folding I-beam introduced in Dyalog 15.

`TxtToCsv` uses the file-system primitives `⎕NINFO`, `⎕NGET`, and `⎕NPUT` introduced in Dyalog 15.

`TxtToCsv` observes *Cannon's Canon*, which tells us to avoid representing options with numeric constants. Instead, it assigns descriptive names at the start and uses those. That's not too bad, but if every function did the same, we risk multiple definitions of the same constant, breaching the DRY principle – don't repeat yourself. (And open to errors: *A man with two watches never knows what time it is.*) We can do better by defining all the Dyalog constants we want to use in a single namespace in the root.

How to organise the code

To expand this program into distributable software we're going to add features, many of them drawn from the `APLTree` library. To facilitate that we'll first organise the existing code into script files, and write a *build script* to assemble a workspace from them.

Start at the root. We're going to be conservative about defining names in the root of the workspace. Why? Right now the program stands by itself and can do what it likes in the workspace. But in the future your program might become part of a larger APL system. In that case it will share the workspace root with other objects you don't know anything about right now.

So your program will be a self-contained object in the workspace root. Give it a distinctive name, not a generic name such as `Application` or `Root`. From here on we'll call it `MyApp`. (We know: that's almost as bad!)

But there *are* other objects you might define in the root. If you're using classes or namespaces that other systems might also use, define them in the root. For example, if `MyApp` should one day become a module of some larger system, it would make no sense for each module to have its own copy of, say, the `APLTree` class `Logger`.

With this in mind, let's distinguish some categories of code, and how the code in `MyApp` will refer to them.

1. **General utilities and classes** For example, the `APLTreeUtils` namespace and the `Logger` class. (Your program doesn't yet use these utilities.) In the future, other programs, possibly sharing the same workspace, might use them too.
2. **Your program and its modules** Your top-level code will be in `#.MyApp`. Other modules and `MyApp`-specific classes may be defined within it.
3. **Tools and utility functions specific to MyApp** These might include your own extensions to Dyalog namespaces or classes. Define them inside the app object, eg `#.MyApp.Utils`.
4. **Your own language extensions and syntax sweeteners** For example, you might like to use functions `means` and `else` as simple conditionals. These are effectively your local *extensions* to APL, the functions you expect always to be around. Define your collection of such functions into a namespace in the root, eg `#.Utilities`.

The object tree in the workspace might eventually look something like:

```
#
|-⊗Constants
|-⊗APLTreeUtils
|-⊗Utilities
|-○MyApp
|  |-⊗Common
|  |-⊗Engine
|  |-○TaskQueue
|  \-⊗UI
\--○Logger
```



⊗ denotes a namespace, ○ a class.

The objects in the root are ‘public’. They comprise `MyApp` and objects other applications might use. (You might add another application that uses `#.Utilities`. Everything else is encapsulated within `MyApp`. Here’s how to refer in the `MyApp` code to these different categories of object.

1. `log←NEW #.Logger`
2. `queue←NEW TaskQueue`
3. `tbl←Engine.CountLetters txt`
4. `r←ok,ok #.Utilities.means r #.Utilities.else 'error'`

That last is pretty horrible, but we can improve it by defining aliases within `#.MyApp`:

```
(C U)←#.(Constants Utilities)
```

allowing (4) to be written as

```
r←ok,ok U.means r U.else 'error'
```

Why not use `⊞PATH`?

`⊞PATH` tempts us. We could set `⊞PATH←'#.Utilities'`. The expression above could then take its most readable form:

```
r←ok,ok means r else 'error'
```

Trying to resolve the names `means` and `else`, the interpreter would consult `⌈PATH` and find them in `#.Utilities`. So far so good: this is what `⌈PATH` is designed for. It works fine in simple cases, but it will lead us into problems later:

- As long as each name leads unambiguously to an object, shift-clicking on it will display it in the editor, a valuable feature of APL in development and debugging. The editor allows us to change code during execution, and save those changes back to the scripts. But the editor doesn't consult `⌈PATH`, so setting it breaks that valuable connection.
- Understanding the scope of the space in which a GUI callback executes can be challenging enough; introducing `⌈PATH` makes it harder still.

Project Gutenberg

We'll raid [Project Gutenberg](https://www.gutenberg.org/)¹ for some texts to read.

We're tempted by the complete works of William Shakespeare but we don't know that letter distribution stayed constant over four centuries. Interesting to find out, though, so we'll save a copy as `Z:\texts\en\shakespeare.dat`. And we'll download some 20th-century books as TXTs into the same folder. Here are some texts we can use.

```
↑>(⌈NINFO⌈1) 'z:\texts\en\*.txt'
z:/texts/en/ageofinnocence.txt
z:/texts/en/dubliners.txt
z:/texts/en/heartofdarkness.txt
z:/texts/en/metamorphosis.txt
z:/texts/en/pygmalion.txt
z:/texts/en/timemachine.txt
z:/texts/en/ulysses.txt
z:/texts/en/withthesehands.txt
z:/texts/en/wizardoz.txt
```

MyApp reloaded

We'll first make MyApp a simple 'engine' that does not interact with the user. Many applications have functions like this at their core. Let's enable the user to call this engine from the command line with appropriate parameters. By the time we give it a user interface, it will already have important capabilities, such as logging errors and recovering from crashes.

¹<https://www.gutenberg.org/>

Our engine will be based on the `TxtToCsv` function. It will take one parameter, a fully-qualified filepath for a folder or file. If a file, it will write a sibling CSV. If a folder, it will read any TXT files in the folder, count the letter frequencies and report them as a CSV file sibling to the folder. Simple enough. Here we go.

Building from a DYAPP

In your text editor open a new document. (Or you can take the DYAPP from the folder for this chapter.)



You need a text editor that handles Unicode. If you're not already using a Unicode text editor, Windows' own Notepad will do for occasional use. (Set the default font to APL385 Unicode.) For a full-strength multifile text editor [Notepad++²](https://notepad-plus-plus.org/) works well.

Here's how the object tree will look:

```
#
|-*Constants
|-*Utilities
\-*MyApp
```



See the *SALT User Guide* for more about DYAPPs.

The file tree will look like this:

```
z:\code\v01\Constants.dyalog
z:\code\v01\Utilities.dyalog
z:\code\v01\MyApp.dyapp
```

So `z:\code\v00\MyApp.dyapp` looks like this:

²<https://notepad-plus-plus.org/>


```
Target #  
Load Constants  
Load Utilities  
Load MyApp
```

and the DIALOGs look like this. `Constants.dyalog`:



You can download all the scripts in this chapter from the corresponding folder in the book website. Or create the namespaces in the session and use SALT to save them to files.

```
:Namespace Constants  
  
  A Dyalog constants  
  
  :Namespace NINFO  
  
    WILDCARD←1  
  
  :EndNamespace  
  
  :Namespace NPUT  
  
    OVERWRITE←1  
  
  :EndNamespace  
  
:EndNamespace
```

Later on we'll introduce a more convenient way to represent and maintain the definitions of constants. This will do nicely for now.

And `Utilities.dyalog`:

:Namespace Utilities

```

  A Ubiquitous functions that for local purposes
  A effectively extend the language
  A Treat as reserved words: do not shadow

```

```

caseDn←{0(819I)ω}
caseUp←{1(819I)ω}

map←{
  (old new)←α
  nw←uω
  (new,nw)[(old,nw)ιω]
}

```

:EndNamespace

And another:

:Namespace MyApp**A Aliases**

```

(C U)←#.(Constants Utilities) A must be already defined

```

```

A === VARIABLES ===

```

```

ACCENTS←↑'ÁÂÃÄÅÇÐÈÊËÉÌÍÎÏÑÒÓÔÕÖØÙÚÛÜÝ' 'AAAAAACDEEEEEIIIIINOOOOOOUUUUUY'

```

```

A === End of variables definition ===

```

```

CountLetters←{
  {α(≠ω)}⊔A{ω≠~ω∈α}(↑ACCENTS)U.map U.caseUp ω
} A Table of letter counts in string ω

```

```

▽ {ok}←TxtToCsv fullfilepath;xxx;csv;stem;path;files;txt;type;lines;nl;enc;t\
gt;src;tbl

```

```

  A Write a sibling CSV of the TXT located at fullfilepath,
  A containing a frequency count of the letters in the file text
  csv←'.csv'
  :Select type←1 □NINFO fullfilepath
  :Case 1 A folder

```

```

        tgt←fullfilepath, csv
        files←(⊖NINFO⊖C.NINFO.WILDCARD)fullfilepath, '\*.txt'
:Case 2 A file
    (path stem xxx)←⊖NPARTS fullfilepath
    tgt←path, stem, csv
    files←, <fullfilepath
:EndSelect
tbl←0 2ρ'A' 0
:For src :In files
    (txt enc nl)←⊖NGET src
    tbl←CountLetters txt
:EndFor
lines←{α, ', ', ⚡ω}/>{α(+/ω)}⊖/↓[1]tbl
ok←×(lines enc nl)⊖NPUT tgt C.NPUT.OVERWRITE
▽

```

```
:EndNamespace
```

Launch the DYAPP by double-clicking on its icon in Windows Explorer. Examine the active session. We see

```

- Constants
    - NINFO
        - WILDCARD
    - NPUT
        - OVERWRITE
- LocalAPL
    - caseDn
    - caseUp
    - map
- MyApp
    - ACCENTS
    - CountLetters
    - TxtToCsv

```



If you also see containers `SALT_Data` ignore them. They are part of how the Dyalog editor updates script files.

We have converted the saved workspace to a DYAPP that assembles the workspace from DYALOGs. We can use `MyApp.dyapp` anytime to recreate the app as a workspace. But we have not saved a workspace. We will always assemble a workspace from scripts.

Package MyApp as an executable

Output to the session log

For Version 1.0 we'll package `MyApp` as an EXE. It's going to run 'headless' – without a user interface (UI). It won't have a session. What happens to values that would otherwise be written in the session log? They disappear. That's not actually a problem, but it is tidy to catch anything that would otherwise be written to the UI, including empty arrays.

`TxtToCsv` has a shy result, so it won't write its result to the session. That's fine.

We'll also fix three key environment variables for it in `MyApp`:

```
(⊞IO ⊞ML ⊞WX)←1 1 3 ⌈ environment variables
```

Reading arguments from the command line

`TxtToCsv` needs an argument. The EXE must take it from the command line. We'll give `MyApp` a niladic function `StartFromCmdLine`. The DYAPP will use it to start the program:

```
Target #
Load Constants
Load Utilities
Load MyApp
Run MyApp.SetLX
```

and in `MyApp.dyalog`:

```
▽ SetLX
⌈ Set Latent Expression in root ready to export workspace as EXE
#.⊞LX←'MyApp.StartFromCmdLine'
▽

▽ StartFromCmdLine;args
⌈ Read command parameters, run the application
  {}TxtToCsv 2>2↑⊞2 ⊞NQ'.' 'getcommandlineargs'
▽
```

This is how MyApp will run when called from the Windows command line.

We're now nearly ready to export the first version of our EXE.

1. From the File menu pick *Export*.
2. Pick `Z:\` as the destination folder.
3. From the list *Save as type* pick *Standalone Executable*.
4. Set the *File name* as **MyApp**.
5. Check the *Runtime application* and *Console application* boxes.
6. Click *Save*.

You should see an alert message: *File Z:MyApp.exe successfully created*.



Use the *Version* button to bind to the EXE information about the application, author, version, copyright and so on. Specify an icon file to replace the Dyalog icon with one of your own.

Let's run it. From a command line:

```
c:\Users\A.N. Other>CD Z:\  
Z:\>MyApp.exe texts\en
```

Looking in Windows Explorer at `Z:\texts\en.csv`, we see its timestamp just changed. Our EXE works!

Logging what happens

MyApp 1.0 is now working, but handles errors poorly. See what happens when we try to work on a non-existent folder. In the command shell

```
Z:\>MyApp.exe texts\de
```

We see an alert message: *This Dyalog APL runtime application has attempted to use the APL session and will therefore be closed.*

MyApp failed, because there is no folder `Z:\texts\de`. That triggered an error in the APL code. The interpreter tried to signal the error to the session. But a runtime task has no session, so at that point the interpreter popped the alert message and MyApp died.

MyApp 2.0 could do better. In several ways.

- Have the program write a log file recording what it's done.
- Set traps to catch and log foreseeable problems.
- Set a top-level trap to catch and report unforeseen errors and save a crash workspace for analysis.

Save a copy of `Z:\code\v01` as `Z:\code\v02`.

Start with the log file. We'll use the `APLTree Logger` class, which we'll now install in the workspace root. If you've not already done so, copy the `APLTree` library folder into `Z:\code`. Now edit `Z:\code\v02\MyApp.dyapp` to include some library code:

```
Target #
Load ..\AplTree\APLTreeUtils
Load ..\AplTree\Logger
Load ..\AplTree\WinFile
Load Constants
Load Utilities
Load MyApp
Run MyApp.SetLX
```

and use the `DYAPP` to recreate the `MyApp` workspace.

The `Logger` class is now part of `MyApp`. Let's get the program to log what it's doing.

Within `MyApp`, some changes. Some aliases for the new code.

```

A Aliases (referents must be defined previously)
  (A L W)←#.(APLTreeUtils Logger WinFile) A from APLTree
  (C U)←#.(Constants Utilities)

```

We'll write logfiles into a subfolder of (wherever) the EXE lives. So we need `TxtToCsv` to ensure this subfolder exists. To do that, we'll first set the Current Directory to the EXE's container.

```

▽ SetLX
A Set Latent Expression in root ready to export workspace as EXE
  W.PolishCurrentDir A set current dir to that of EXE
  #.□LX←'MyApp.StartFromCmdLine'
▽

```

Now that `TxtToCsv` can log what it's doing, it makes sense to check its argument. We wrap the earlier version of the function in an if/else:

```

▽ {ok}←TxtToCsv fullfilepath;Δ;xxx;csv;stem;path;files;txt;type;lines;nl;enc\
;tgt;src;tbl
  A Write a sibling CSV of the TXT located at fullfilepath,
  A containing a frequency count of the letters in the file text
  'CREATE!'W.CheckPath'Logs' A ensure subfolder of current dir
  Δ←L.CreatePropertySpace
  Δ.path←'Logs\' A subfolder of current directory
  Δ.encoding←'UTF8'
  Δ.filenamePrefix←'MyApp'
  Δ.refToUtils←#
  Log←NEW L(,Δ)
  Log.Log'Started MyApp in ',W.PWD
  Log.Log'Source: ',fullfilepath

  :If ~□NEXISTS fullfilepath
    Log.Log'Invalid source'
    ok←0
  :Else

    A as before...

  :EndIf
▽

```

Notice how defining the aliases `A`, `L`, and `W` in the namespace script – the environment of `StartFromCmdLine` and `TxtToCsv` – makes the function code more legible.

The foreseeable error that aborted the runtime task – on an invalid filepath – has now been replaced by log messages to say no files were found.

Where to keep the logfiles?

Where is MyApp to write the logfile? We need a filepath we know exists. That rules out `fullfilepath`. (We need a logfile even if that isn't a valid path.) But we do know the EXE exists, or MyApp would not be running. (*Executo ergo sum.*)

`StartFromCmdLine` uses `W.PolishCurrentDir` to set the current directory to that of the EXE. `TxtToCsv` then uses `W.CheckPath` to ensure the current directory contains a folder `Logs`. That can then be safely specified as the `path` property for the `Logger` instance.

Why not do both in `TxtToCsv`? Because we might, while developing in the session, want our `CurrentDirectory` elsewhere. In this case, when we run the DYAPP, the `Logs` folder will be a sibling of the DYAPP, handily producing a `Logs` folder for each version of the project.

As the logging starts and ends in `TxtToCsv`, we can run this in the workspace now to test it.

```
)CS
#
  #.WinFile.Cd 'Z:\' a working directory
C:\Windows\system32
  #.WinFile.PWD a really?
Z:\
  a yes
  #.MyApp.TxtToCsv 'Z:\texts\en'
```

Let's see what we got.

```
  #.WinFile.PWD a still here?
Z:\
  =>(NINFORM1) 'Logs\*.LOG'
MyApp_20160406.log
  ↑NGET 'Logs\MyApp_20160406.log'
2016-04-06 13:42:43 *** Log File opened
2016-04-06 13:42:43 (0) Started MyApp in Z:\code\v02
2016-04-06 13:42:43 (0) Source: Z:\texts\en
2016-04-06 13:42:43 (0) Target: Z:\texts\en.csv
2016-04-06 13:42:43 (0) 244 bytes written to Z:\texts\en.csv
2016-04-06 13:42:43 (0) All done
```

Let's see if this works also for the exported EXE. Run the DYAPP to rebuild the workspace. Export as before, and in the Windows command line run the new `MyApp.exe`.


```
Z:\>MyApp.exe texts\en
```

Yes! The output TXT gets produced as before, and the work gets logged in Z:\Logs.

Let's see what happens now when the filepath is invalid.

```
Z:\>MyApp.exe texts\de
```

No warning message – the program made an orderly finish. And the log?

```
↑[NGET 'Logs\MyApp_20160406.log'
2016-04-06 13:42:43 *** Log File opened
2016-04-06 13:42:43 (0) Started MyApp in Z:\code\v02
2016-04-06 13:42:43 (0) Source: Z:\texts\en
2016-04-06 13:42:43 (0) Target: Z:\texts\en.csv
2016-04-06 13:42:43 (0) 244 bytes written to Z:\texts\en.csv
2016-04-06 13:42:43 (0) All done
2016-04-06 13:42:50 *** Log File opened
2016-04-06 13:42:50 (0) Started MyApp in Z:\code\v02
2016-04-06 13:42:50 (0) Source: Z:\texts\de
2016-04-06 13:42:50 (0) Invalid source
```

Yes!

We now have MyApp logging its work in a subfolder of the application folder and reporting problems which it has anticipated.

Next we need to consider how to handle and report errors we have *not* anticipated. We should also return some kind of error code to Windows. If MyApp encounters an error, any process calling it needs to know.

Handling errors

Our app now anticipates, tests for and reports certain foreseeable problems with the parameters. We'll now handle errors more comprehensively.

What are we missing?

1. Other problems are foreseeable. The file system is a rich source of ephemeral problems and displays. Many of these are caught and handled by the APLTree utilities. They might make several attempts to read or write a file before giving up and signalling an error. Hooray. We need to handle the events signalled when the utilities give up.
2. The MyApp EXE terminates with an all-OK zero exit code even when it has caught and handled an error. Some custom exit codes would make it a better Windows citizen.
3. By definition, unforeseeable problems can't be foreseen. But we foresee there will be some! A mere typo in the code could break execution. We need a master trap to catch any events that would break execution, save them for analysis, and report them in an orderly way.

Foreseen errors

We'll start with the first two. Quite a bit of restructuring here. We'll get `Compile` to return an exit code to be used by `OFF`. So we'll begin by defining in the `MyApp` namespace a reference namespace of constants and their exit codes.

```
:Namespace MyApp
  ⍵IO←1 ⍵ML←1 ⍵WX←3
  (A L W)←#.(APLTreeUtils Logger WinFile) A aliases for all

  A define Exit Code constants
  EXIT←⍵NS''
  EXIT.OK←0
  EXIT.INVALID_SOURCE_FOLDER←101
  EXIT.INVALID_TARGET_FOLDER←102
  EXIT.UNABLE_TO_READ_MANIFEST←103
  EXIT.NO_FILES_FOUND←104
  EXIT.UNABLE_TO_READ_SOURCE←105
  EXIT.UNABLE_TO_WRITE_TO_TARGET←106
```

Note that we define an OK value of zero for completeness. All the exit codes are defined here. The function code can refer to them by name, so the meaning is clear. And this is the only definition of the exit-code values.

```

▽ StartFromCmdLine
  A Initialise environment, read command parameters, and run the application
  W.PolishCurrentDir A set current dir to that of EXE
  □OFF Compile 1+3↑(□2 □NQ'.' 'getcommandlineargs'),' ' ' '
▽

```

No change here.

`Compile` still starts and stops the logging, but it now calls `CheckAgenda` to examine the agenda (things to be done) and `CompileFiles` to do them.

```

▽ exit←Compile(srcfolder tgtfile);Δ;Log;Error;files
  A Dyalog Cookbook compiler
  A Compile files from srcfolder to tgtfile
  'CREATE!'W.CheckPath'Logs' A ensure subfolder of current dir
  Δ←L.CreatePropertySpace
  Δ.path←'Logs\' A subfolder of current directory
  Δ.encoding←'UTF8'
  Δ.filenamePrefix←'MyApp'
  Δ.refToUtils←#
  Log←□NEW L(,Δ)
  Log.Log'Started MyApp in ',Δ.path
  Log.Log'Source: ' 'Target: ',srcfolder tgtfile

  Error←Log°{code←EXIT±ω ◊ code←α.LogError code ω}

  :If 0⇒(exit files)←CheckAgenda srcfolder tgtfile
    exit←CompileFiles files tgtfile
  :EndIf

  Log.Log'All done'
  Log.Close
▽

```

Some comments

- **Error:** this direct function aids clarity by avoiding some repetition in `CheckAgenda` and `CompileFiles`, where one would otherwise write perhaps: `exit←EXIT.INVALID_SOURCE_FOLDER Log.LogError exit 'INVALID_SOURCE_FOLDER'`

`Compile` now uses `CheckAgenda` to confirm it has work to do.

```

    ▽ r←CheckAgenda(srcfolder tgtfile);Δ;_srcfolder;orbust;manifest;files;srcfil\
e;exit;n
    A Validate and set up agenda
    A r: errorcode filelist
    A errorcode: (int) see EXIT
    A filelist: (strs) full filepaths
    _srcfolder←{ω,(~'\/'ε~>φω)/'\'}srcfolder
    orbust←{11::0 ♦ αα ω} A return 0 if test breaks
    exit←EXIT.OK ♦ files←'' A default values

    :If ~W.DoesExistDir orbust _srcfolder
        exit←Error'INVALID_SOURCE_FOLDER'
    :ElseIf ~W.DoesExistDir orbust W.Parent tgtfile
        exit←Error'INVALID_TARGET_FOLDER'
    :Else
        manifest←_srcfolder,'MANIFEST.DAT'
        :If W.DoesExistFile orbust manifest
            :Trap 0
                files←A.ReadUtf8File manifest
                Log.Log(⌘#files),' files specified in manifest'
            :Else
                exit←Error'UNABLE_TO_READ_MANIFEST'
            :EndTrap
        :Else
            files←W.Dir _srcfolder,'*.TXT'
        :EndIf

        :If exit=EXIT.OK A let's see what we got...
            files,~''←c_srcfolder A full filepaths
            files/~←W.DoesExistFile files
            :If xn←#files
                Log.Log(⌘n),' file',( 's'/~n>1),' found in folder'
            :Else
                exit←Error'NO_FILES_FOUND'
            :EndIf
        :EndIf

    :EndIf

    r←exit files
    ▽

```

Some comments

- Testing and logging the parameters has swollen this part of the code to where we're pleased to have it in its own function. It also defines `files` and is the last code to read `srcfolder`, so this is a good functional encapsulation.
- `_srcfolder`: suffixing a folder path with `\` is so common we might define a `suffix` function to do it. But the definition is barely twice the length of the name we would give it!
- The `αα` in `orbust` marks it as an operator, modifying how a function works. `αα` refers to the function. The error guard `0::0` means in the event of any error return 0 (false). For example, `W.DoesExistDir` signals 11 (domain error) if its argument contains a wildcard. Moderated by `orbust`, it returns 0.
- `files,~''←_srcfolder` combines the *commute* and *each* operators with assignment through a function to prefix each filename with `_srcfolder`. Assignment through a function elides the left argument of a dyadic function, so `file←_srcfolder` is equivalent to `file←file,_srcfolder`. But we want to prefix `file` with `_srcfolder`, not suffix it. So we use *commute* to switch the arguments of *catenate*: `file,~''←_srcfolder`, equivalent to `file←_srcfolder,file`. Finally, we want to prefix every file in `files` so the assignment is done through `,~''` and the right argument enclosed. Similarly `files/~''←` compresses `files` with a Boolean vector.
- Note the exit code is tested `exit=EXIT.OK`. Testing `~*exit` would work and read as well, but relies on `EXIT.OK` being 0. The point of defining the codes in `EXIT` was to make the functions relate to the exit codes only by their names.
- In general, we like functions to start at the top and exit at the bottom. The best form is a cascade of tests:

```
:If      ~ok←test1 ♦ msg←'Failed test 1'
:ElseIf ~ok←test2 ♦ msg←'Failed test 2'
:ElseIf ~ok←test3 ♦ msg←'Failed test 3'
:Else
      (ok msg)←DoAllThe Work
:EndIf
r←ok msg
```

The handling of error codes and messages can easily obscure the rest of the logic. Clarity is not always easy to find, but is well worth working for. This is particularly true where there is no convenient test for an error, only a trap for when it is encountered.

In such cases, it is tempting to use a `:Return` statement to abort the function. It can be confusing when a function 'aborts' in the middle. We have learned a great respect for our capacity to get confused. Aborting from the middle of a function may also skip essential tidying up at the end.

We meet the same issue in `CompileFiles`, where we trap errors and abort within a loop. Note how the use of `repeat/until` allows – unlike a `for` loop – to test at the bottom of the loop both the counter and the exit code.

```

▽ exit←CompileFiles(srcfiles tgtfile);srcfile;txt;i
A Copy srcfiles to tgtfile
A exit: (int) see EXIT
A srcfiles: (strs) full filepaths
A tgtfile: (str) full filepath of target file
W.Delete tgtfile
(exit i)←EXIT.OK □IO
:Repeat
    srcfile←i>srcfiles
    :Trap 0
        txt←'flat'A.ReadUtf8File srcfile
    :Else
        exit←Error'UNABLE_TO_READ_SOURCE'
    :EndTrap
    :If exit=EXIT.OK
        Log.Log'Read ',(i>txt),' bytes from ',srcfile
        :Trap 0
            'append'A.WriteUtf8File tgtfile txt
        :Else
            exit←Error'UNABLE_TO_WRITE_TO_TARGETFILE'
        :EndTrap
    :EndIf
    i←i+1
:Until i>#srcfiles
:OrIf exit≠EXIT.OK
▽

```

Some comments

- :Trap 0: trap any error

Unforeseen errors

Our code so far covers the errors we foresee: errors in the parameters, and errors encountered in the file system. There remain the unforeseen errors, chief among them errors in our own code. If the code we have so far breaks, the EXE will try to report the problem to the session, find no session, and abort with an exit code of 4 to tell Windows “Sorry, it didn’t work out.”

If the error is easily replicable, we can easily track it down using the development interpreter. But the error might not be easily replicable. It could, for instance, have been produced by ephemeral congestion on a network interfering with file operations. Or the parameters for your app might be

so complicated that it is hard to replicate the environment and data with confidence. What you really want for analysing the crash is a crash workspace, a picture of the ship before it went down.

For this we need a high-level trap to catch any event not trapped by `Compile`. We want it to save the workspace for analysis. We might also want it to report the incident to the developer: users don't always do this. For this we'll use the `HandleError` class from the APLTree.

Place into `C:\dev\Myapp\src\` a copy of `HandleError.dyalog` from the APLTree. Edit `C:\dev\MyApp\MyApp.d`

```
Target #
Load C:\dev\MyApp\src\APLTreeUtils
Load C:\dev\MyApp\src\WinFile
Load C:\dev\MyApp\src\HandleError
Load C:\dev\MyApp\src\Logger
Load C:\dev\MyApp\src\MyApp
```

And set an alias `H` for it in the preamble of the `MyApp` namespace:

```
:Namespace MyApp
  ⍵IO←1 ⍵ML←1 ⍵WX←3
  (A H L W)←#.(APLTreeUtils HandleError Logger WinFile)

  ⍝ define Exit Code constants
  EXIT←⍵NS''
  EXIT.OK←0
  EXIT.APPLICATION_CRASHED←100
  EXIT.INVALID_SOURCE_FOLDER←101
  EXIT.INVALID_TARGET_FOLDER←102
  EXIT.UNABLE_TO_READ_MANIFEST←103
  EXIT.NO_FILES_FOUND←104
  EXIT.UNABLE_TO_READ_SOURCE←105
  EXIT.UNABLE_TO_WRITE_TO_TARGET←106
```

Note the new exit code to indicate the application crashed.

We'll start as soon as `StartFromCmdLine` begins, setting `HandleError` to do whatever it can before we can give it more specific information derived from the calling environment.

```

▽ exit←StartFromCmdLine args
A Initialise environment, and run the application
A args: command line arguments
A exit: Windows exit code -- see EXIT
W.PolishCurrentDir A set current dir to that of EXE
□TRAP←0 'E' '#.HandleError.Process '''' A trap unforeseen problems
exit←Compile(args,'')[2 3]
▽

```

This trap will do to get things started and catch anything that falls over immediately. We need to get more specific in `Compile`.

```

▽ exit←Compile(srcfolder tgtfile);Δ;Log;Error;files
A Dyalog Cookbook compiler
A Compile files from srcfolder to tgtfile

'CREATE!'W.CheckPath'Logs' A ensure subfolder of current dir
Δ←L.CreatePropertySpace
Δ.path←'Logs\' A subfolder of current directory
Δ.encoding←'UTF8'
Δ.filenamePrefix←'MyApp'
Δ.refToUtils←#
Log←NEW L(,Δ)
Log.Log'Started MyApp in ',Δ.path
Log.Log'Source: ' 'Target: ',srcfolder tgtfile
Error←Log{code←EXIT⊘ code←α.LogError code ω}

A refine trap definition
#.ErrorParms←H.CreateParms
#.ErrorParms.errorFolder←W.PWD
#.ErrorParms.returnValue←EXIT.APPLICATION_CRASHED
#.ErrorParms.(logFunctionParent logFunction)←Log'Log'
#.ErrorParms.trapInternalErrors←~isDev
mayBreak←'Development'≡4>'.'WG'APLVersion'
means←{αφω} ⊘ else←{ω α}
□TRAP←mayBreak means(0ρ□TRAP)else 0 'E' '#.HandleError.Process ''#.ErrorPa\
rms'''

:If 0⇒(exit files)←CheckAgenda srcfolder tgtfile
exit←CompileFiles files tgtfile
:EndIf

```



```

    Log.Log 'All done'
    Log.Close
▽

```

`CheckAgenda` and `CompileFiles` are as before.

Some comments

- `mayBreak` flags whether (true) the application is allowed to suspend, or (false) errors are to be caught and handled. Later we will set this flag in an INI file. For now we set it by testing whether the interpreter is development or runtime.
- `means ... else` are 'syntax sweeteners' for a simple conditional. A bit over the top to define them merely to use them on the next line. Later we shall look at how you can incorporate functions such as these into your development environment as your local extensions to the APL language. Watch Out: unlike with control structures, both expressions get evaluated, regardless of the value of `test`.
- `⌈TRAP` – if we're using a Development interpreter, any error will simply suspend – including errors inside `HandleError` itself. Note the form `⌈test⌋expn1 expn2` as a simple conditional construct.
- `#.ErrorParms.errorFolder` – write crash files as siblings of the EXE.
- `#.ErrorParms.(logFunctionParent logFunction)` – we set a ref to the `Logger` instance, so `HandleError` can write on the log.

We can test this! Put a `stop`³ in `CompileFiles` after copying the first file to the target.

```

        :EndIf
        i←i+1
        . A DEBUG
    :Until i>#srcfiles
    :OrIf exit≠EXIT.OK
▽

```

This will definitely break. It is not caught by any of the other traps. Export the workspace as before, run it from the DOS command shell... and what do we get?

Predictably we get a new TXT, with only the first file copied to it. In `C:\dev`, we find in the LOG a record of what happened.

First, the log entry records the crash then breaks off:

³The poets among us love that the tersest way to bring a function to a full stop is to type one.

```
2014-12-21 15:42:43 *** Log File opened
2014-12-21 15:42:43 (0) Started C:\dev\myapp
2014-12-21 15:42:43 (0) Source: temp\source
2014-12-21 15:42:43 (0) Target: temp\test.txt
2014-12-21 15:42:43 (0) 2 files specified in manifest
2014-12-21 15:42:43 (0) 2 files found
2014-12-21 15:42:43 (0) Read 45 bytes from temp\source\foo.txt
2014-12-21 15:42:43 (0) *** Error
2014-12-21 15:42:43 (0) Error number=2
2014-12-21 15:42:43 (0) SYNTAX ERROR
2014-12-21 15:42:43 (0) CompileFiles[17] . A DEBUG
2014-12-21 15:42:43 (0) ^
```

We also have an HTM with a crash report, an eponymous DWS containing the workspace saved at the time it broke, and an eponymous DCF whose single component is a namespace of all the variables defined in the workspace. Some of this has got to help.

Remove the deliberate error from `#.MyApp` and save your work.

Discussion

[SJT] What's a convenient way to show the exit codes returned to the Windows command shell?

Offcuts

Initialising with INI files

Testing: the sound of breaking glass

Our app here is as simple as we could make it – just join a folder of text files into a single TXT. For many purposes, simply tying all the source files to a vector of tie numbers `srcs` and creating a new file on tie number `tgt` would reduce the app to a single line:

```
(fread(srcs,82,NSIZE,srcs) fappend tgt
```

Pretty much all the other code has been written to package that ‘app’ for shipment and to control how it behaves when it encounters problems.

Developing code refines and extends it. We have more developing to do. Some of that developing is liable to break what we already have working. Too bad. No one’s perfect. But we would at least like to know when we’ve broken something – to hear the sound of breaking glass behind us. Then we can fix the error before going any further.

In our ideal world we would have a team of testers continually testing and retesting our latest build to see if it still does what it’s supposed to do. The testers would tell us if we broke anything. In the real world we have programs – tests – to do that.

What should we write tests for? “Anything you think might break,” says Kent Beck⁴, author of *Extreme Programming Explained*. We’ve written code to allow for ways in which the file system might surprise us. We should write tests to discover if that code works. We’ll eventually discover conditions we haven’t foreseen and write fixes for them. Now those conditions too join the things we think might break, and get added to the test suite.

⁴in conversation with one of the authors.

Testing in different versions of Windows

User interface – native

A simple UI with native Dyalog forms

Forms

Controls

Callbacks and the event queue

Extended controls

Providing help

(Choice between CHM generated by say *Help and Manual* or the APLTreeHelp namespace.)

Writing an installer

Working with other processes

Launching tasks

Running as a Windows service

Sharing files

Communicating through TCP/IP

Managing your source code

Documentation

FIRE

CompareSimple

Useful user commands

Storing data

Native and component files

XML

JSON

Relational databases

User interface – WPF

User interface – HTTP

Deploying as a Web application

Deploying as a Web service

2. Professional programming practices

Stay DRY – don't repeat yourself

If you say something twice (or more) and later have to change it, you have to change it everywhere you said it. Here are some refactoring techniques for removing duplications from your code.

APL.local

Every developer, and every development group, has utility functions uses throughout the code base. Most of these can and should be grouped into topic-specific namespaces, such as the `WinFile` namespace from the `APLTree` library. Aliasing the namespaces, such as `W←#.WinFile` allows utility functions to be called in abbreviated form, e.g. `W.Dir` instead of `#.WinFile.Dir`.

But some functions are so ubiquitous and general it makes better sense to treat them as your local extensions to the language itself. For example the simplified conditional functions `means` and `else` are ‘syntax sweeteners’ that allow you to write

```
:if a=b
    Z←'this'
:else
    Z←'that'
:endif
```

more legibly as `Z←(a=b) means 'this' else 'that'`

Functions `means` and `else` could be defined in say a namespace `#.Utilities` and abbreviated to `U`, permitting

```
Z←(a=b) U.means 'this' U.else 'that'
```

But you might reasonably prefer to omit even the `U.` prefix.

Defining ubiquitous utilities

Here is how to define your `Utilities` namespace in the workspace root so that you can refer to them without prefixes.



Every function or object you include in the root as your ‘local extension’ to the APL language effectively becomes a reserved word in your ‘local’ APL dialect. Be conservative and define functions this way only when you have found them ubiquitous and indispensable!

...

Some utilities you might like to make ubiquitous

Hooray for arrays

You've already discovered how APL lets you write code that is helpfully light on 'ceremony'. For example, you don't have to declare variable types or loop through collections. Here are some more advanced array programming techniques that might give you a further lift.

Functional is funky

Functional style makes your code easier to test, debug and re-use.

Complex data structures

Some ways array programmers commonly use to represent more complex data structures, such as trees, dictionaries and tables.

Passing parameters perfectly

The ghastliness of globals

Nothing is handier when developing your code than keeping bits of information in global variables. They are like the Post-It™ notes or scraps of paper on your real-world desktop. They have no place in your application. Here's why professional programmers keep the global symbol table as empty as possible – and how.

Coding with class

Classes are like micro-workspaces, and a great way of organising your code and data into modules.